Renormalization of SI-2PIEA gap equations in the Hartree-Fock approximation

Author: Michael Brown
Supplement to thesis Chapter 3

Mathematica notebook to compute couter-terms for two loop truncations of the two particle irreducible effective action

```
ClearAll[veom, geom, neom, divergentpartrules, mg2soln, cteq, cts, δm, δλ];
```

Hartree-Fock gap equations with counterterms

Goldstone equation of motion. Quantities in reference to the thesis are:
p is the four-momentum flowing through the propagators $\Delta_G{}^{-1}$ and $\Delta_N{}^{-1}$,
mg2 is the Goldstone mass squared $m_G{}^2$,
mn2 is the Higgs mass squared $m_H{}^2$,
Z and ZΔ are the wavefunction a propagator renormalization constants,
m^2 is the (renormalized) Lagrangian mass parameter, $\delta m_0{}^2$, $\delta m_1{}^2$ are its counter-terms,
λ is the (renormalized) four point coupling,
$\delta\lambda_0$, $\delta\lambda_{1\,a}$, $\delta\lambda_{1\,b}$, $\delta\lambda_{2\,a}$, $\delta\lambda_{2\,b}$ are the independent coupling counter-terms,
v is the scalar field vacuum expectation value,
$\hbar$ is the reduced Planck constant,
n is the number of fields in the O(n) symmetry group,
t∞g, t∞n are the divergent tadpole integrals for the Goldstone, Higgs resp.,
tfing, tfinn are the finite parts of the tadpoles for the Goldstone, Higgs resp.

Equations of motion

Vev equation of motion

$$\text{(*veom=}$$
$$Z\Delta^{-1}\left(m^2+\delta m_0{}^2\right)v+\frac{\lambda+\delta\lambda_0}{6}v^3+\frac{\hbar}{6}Z\Delta\left(n-1\right)(\lambda+\delta\lambda_{1a})\,v\,(t\infty g+tfing)+\frac{\hbar}{6}Z\Delta\left(3\lambda+\delta\lambda_{1a}+2\delta\lambda_{1b}\right)v\,(t\infty n+tfinn)$$
$$\text{finveom}=m^2v+\frac{\lambda}{6}v^3+\frac{\hbar}{6}\left(n-1\right)\lambda\ v\ tfing+\frac{\hbar}{2}\lambda\ v\ tfinn\text{*)}$$

```
veom = v mg2
```

Goldstone equation of motion

$$\text{geom} = p^2 - \text{mg2} == Z\, Z\Delta\, p^2 - m^2 - \delta m_1{}^2 - Z\Delta \frac{\lambda + \delta\lambda_{1\,a}}{6} v^2 - \frac{\hbar}{6} \left(\left(n+1\right) \lambda + \left(n-1\right) \delta\lambda_{2\,a} + 2\, \delta\lambda_{2\,b}\right) Z\Delta^2 \, (\text{t}\infty\text{g} + \text{tfing}) - \frac{\hbar}{6} (\lambda + \delta\lambda_{2\,a})\, Z\Delta^2\, (\text{t}\infty\text{n} + \text{tfinn})$$

$$\text{finmg2} = \text{mg2} \,/.\, \text{Solve}\left[p^2 - \text{mg2} == p^2 - m^2 - \frac{\lambda}{6} v^2 - \frac{\hbar}{6} \left(n+1\right) \lambda\, \text{tfing} - \frac{\hbar}{6} \lambda\, \text{tfinn},\, \text{mg2}\right][[1]]$$

Higgs equation of motion

$$\text{neom} = p^2 - \text{mn2} == Z\, Z\Delta\, p^2 - m^2 - \delta m_1{}^2 - Z\Delta\, v^2 \frac{\left(3\,\lambda + \delta\lambda_{1\,a} + 2\,\delta\lambda_{1\,b}\right)}{6} - \frac{\hbar}{6} (\lambda + \delta\lambda_{2\,a}) \left(n-1\right) Z\Delta^2\, (\text{t}\infty\text{g} + \text{tfing}) - \frac{\hbar}{6} \left(3\,\lambda + \delta\lambda_{2\,a} + 2\,\delta\lambda_{2\,b}\right) Z\Delta^2\, (\text{t}\infty\text{n} + \text{tfinn})$$

$$\text{finmn2} = \text{mn2} \,/.\, \text{Solve}\left[p^2 - \text{mn2} == p^2 - m^2 - v^2 \frac{\lambda}{2} - \frac{\hbar}{6} \lambda \left(n-1\right) \text{tfing} - \frac{\hbar}{2} \lambda\, \text{tfinn},\, \text{mn2}\right][[1]]$$

Infinite parts of tadpoles

c0, c1, Λ and μ are regularisation/renormalisation scheme dependent quantities

$$\text{divergentpartrules} = \left\{\text{t}\infty\text{g} \to \text{c0}\, \Lambda^2 + \text{c1}\, \text{mg2}\, \text{Log}\left[\Lambda^2 / \mu^2\right],\ \text{t}\infty\text{n} \to \text{c0}\, \Lambda^2 + \text{c1}\, \text{mn2}\, \text{Log}\left[\Lambda^2 / \mu^2\right]\right\}$$

Sub in tadpole expressions, eliminate mn2 and solve for mg2

```
mn2fromneom = Solve[neom /. divergentpartrules, mn2][[1]]
```

```
mg2soln = mg2 /. (geom /. divergentpartrules /. mn2fromneom // Solve[#, mg2][[1]] &)
```

```
mn2soln = mn2 /. mn2fromneom /. mg2 → mg2soln // Simplify
```

Gather divergences proportional v, tfing and tfinn and set independently to zero

First we subtract the finite equation of motion, then gather coefficients of the remainder into a list and set each to zero (after some trimming and simplifying).

```
cteq = ((CoefficientList[mg2soln - finmg2, {p, v, tfing, tfinn}] // Flatten) //
        DeleteDuplicates // Simplify // FullSimplify) == 0 // Thread
```

```
cteq2 = ((CoefficientList[mn2soln - finmn2, {p, v, tfing, tfinn}] // Flatten) //
        DeleteDuplicates // Simplify // FullSimplify) == 0 // Thread
```

Solve for counterterms

Find counter-terms from the gap equations

```
cteqs = {cteq, cteq2} // Flatten // FullSimplify // DeleteDuplicates
```

```
cts = {δm1^2, δλ1a, δλ2a, δλ1b, δλ2b, Z, ZΔ} /. Solve[cteqs,
    {δm1, δλ1a, δλ2a, δλ1b, δλ2b, Z, ZΔ}] // FullSimplify // DeleteDuplicates
```

ZΔ is redundant in this truncation, can remove it :

```
cts /. ZΔ → 1 // FullSimplify
```

Verify that the finite gap equations come out right

```
finmg2 ==
  (mg2soln /. Solve[cteqs, {δm1, δλ1a, δλ2a, δλ2b, Z, ZΔ}] /. ZΔ → 1 // FullSimplify //
     DeleteDuplicates) [[2]] // Simplify
```

```
finmn2 == mn2 /.
  ((neom /. divergentpartrules /. mg2 → mg2soln /. Solve[cteqs, {δm1, δλ1a,
          δλ2a, δλ1b, δλ2b, Z, ZΔ}] /. ZΔ → 1 // FullSimplify //
      DeleteDuplicates) // Solve[#, mn2] &) // FullSimplify
```

Verify counter-term expressions in text

```
{δm1^2 == (-ħ λ (n + 2))/6 (c0 Λ^2 + c1 m^2 Log[Λ^2/μ^2]) (δλ1a + λ)/(δλ1b + λ)} /.
     Solve[cteqs, {δm1, δλ1a, δλ2a, δλ1b, δλ2b, Z, ZΔ}] /. ZΔ → 1 //
   FullSimplify // Flatten // DeleteDuplicates
```

```
{δλ1a == δλ2a, δλ1b == δλ2b} /. Solve[cteqs, {δm1, δλ1a, δλ2a, δλ1b, δλ2b, Z, ZΔ}] /.
    ZΔ → 1 // FullSimplify // Flatten // DeleteDuplicates
```

```
{δλ1a / δλ1b} /. Solve[cteqs, {δm1, δλ1a, δλ2a, δλ1b, δλ2b, Z, ZΔ}] /. ZΔ → 1 //
   FullSimplify // Flatten // DeleteDuplicates
```

```
δλ1b /. Solve[cteqs, {δm1, δλ1a, δλ2a, δλ1b, δλ2b, Z, ZΔ}] /. ZΔ → 1 // FullSimplify //
 DeleteDuplicates
```

Total number of independent counter-term equations

```
Length[{cteqs} // Flatten // FullSimplify // DeleteDuplicates] -
 1 (* -1 because one of the "equations" is identically "True" *)
```