Analytic Properties of 2PI Approximation Schemes

Author: Michael Brown

Supplement to thesis Chapter 2

Mathematica notebook to compute solution of the zero dimensional toy model and compare perturbation theory, Pade, Borel-Pade, 2PI and hybrid 2PI-Pade approximations.

Classical action and solutions

```
Scl = 1/2 m^2 q^2 + 1/4! λ q^4;
```

The classical solutions extremize the action:

```
clsoln = Solve[D[Scl, q] == 0, q]
```

The real solution is a minimum and the imaginary solutions are saddle points:

```
D[Scl, q, q] /. clsoln
```

with action:

```
Scl /. clsoln
```

```
ContourPlot[Abs[Exp[-Scl]] /. {m → 1, λ → 1, q → x + ⅈ y}, {x, -3, 3}, {y, -4, 4},
 PlotPoints → 30, Contours → Table[0.5 i, {i, 20}], PlotRange → {0, 10},
 FrameLabel → {"Re[q]", "Im[q]"}, PlotTheme → "Scientific", PlotLegends → Automatic,
 ContourShading → Automatic, ContourStyle → {Black, Dashed}, LabelStyle → Medium]
ContourPlot[Arg[Exp[-Scl]] /. {m → 1, λ → 1, q → x + ⅈ y}, {x, -3, 3},
 {y, -4, 4}, PlotPoints → 30, FrameLabel → {"Re[q]", "Im[q]"},
 Contours → 5, PlotTheme → "Scientific", PlotLegends → Automatic,
 ContourShading → Automatic, ContourStyle → {Black, Dashed}, LabelStyle → Medium]
```

```
(*Export["z-integrand-abs.pdf",%93]
 Export["z-integrand-arg.pdf",%94]*)
```

Partition function and connected generating function

```
ClearAll[normalization, z, w]
```

```
normalization = Integrate[Exp[-Scl], {q, -∞, ∞}, Assumptions → {m > 0, λ > 0}]
```

```
z[k_, l_] := z[k, l] = 1/normalization
  Integrate[Exp[-Scl - 1/2 k q^2 - 1/4! l q^4], {q, -∞, ∞}, Assumptions → {λ + l > 0, m^2 + k > 0}]
```

```
z[k, 0]
```

```
w[k_, l_] := w[k, l] = -Log[z[k, l]]
```

```
w[k, l] // FullSimplify[#, Assumptions → {λ + l > 0, m^2 + k > 0}] &
```

Exact Propagator

```
Gexact = FullSimplify[2 D[w[k, 0], k] /. k → 0]
```

```
Plot[
 {Re[Gexact], Im[Gexact] /. λ → λ - 0.00001 ⅈ, Im[Gexact] /. λ → λ + 0.00001 ⅈ} /. m → 1 //
  Evaluate, {λ, -5, 5}, PlotLegends → {"Re[G̅]", "Im[G̅[λ-ⅈϵ]]", "Im[G̅[λ+ⅈϵ]]"},
 FrameLabel → {λ, G̅}, PlotTheme → "Scientific",
 PlotStyle → {Dashing[{}], Dashed, DotDashed}, LabelStyle → Medium]
```

```
(*Export["Gexact.pdf",%97]*)
```

```
ContourPlot[Abs[Gexact] /. {m → 1, λ → x + ⅈ y}, {x, -3, 3}, {y, -4, 4}, PlotPoints → 50,
 PlotRange → {0, 10}, FrameLabel → {"Re[λ]", "Im[λ]"}, PlotTheme → "Scientific",
 PlotLegends → Automatic, Contours → Table[i, {i, 0, 1.3, 1/10}], LabelStyle → Medium]
```

```
ContourPlot[Arg[Gexact] /. {m → 1, λ → x + ⅈ y}, {x, -3, 3},
 {y, -4, 4}, PlotPoints → 40, FrameLabel → {"Re[λ]", "Im[λ]"},
 PlotTheme → "Scientific", PlotLegends → Automatic, LabelStyle → Medium]
```

```
(*Export["Gexact-abs.pdf",%99]
 Export["Gexact-arg.pdf",%100]*)
```

Discontinuity of the cut in λ:

```
disc = Series[(Gexact /. λ → λ + ⅈ ϵ) - (Gexact /. λ → λ - ⅈ ϵ),
    {ϵ, 0, 0}, Assumptions → {m > 0, λ < 0, ϵ > 0}] // Normal //
  FullSimplify[#, Assumptions → {m > 0, λ < 0, ϵ > 0}] &
```

Spectral function:

```
σexact = (Im[disc] /. λ → -λ)/(-2 π) HeavisideTheta[λ]
```

```
Plot[σexact /. m → 1, {λ, -1, 10}, FrameLabel → {λ, σ},
 PlotTheme → "Scientific", LabelStyle → Medium]
```

```
(*Export["sigma-exact.pdf",%103]*)
```

Exact Vertex Function

```
V4exact =
  Simplify[-(-4 D[w[k, 0], k, k] - 2 Gexact^2)/Gexact^4 /. k → 0, Assumptions → {m > 0, λ > 0}];

Series[V4exact, {λ, 0, 2}]
```

Perturbation series

```
ClearAll[Gpert]

Gpert[n_] := Gpert[n] = Series[Gexact, {λ, 0, n}, Assumptions → {m > 0, λ > 0}] // Normal

Series[Gexact, {λ, 0, 1}, Assumptions → {m > 0, λ > 0}]

Gpert[2]

Table[{n, Gpert[n]}, {n, 20}]; Gpert[20]

Plot[{Re[Gexact], Table[Gpert[n], {n, 0, 5}]} /. m → 1 // Evaluate,
 {λ, -1, 1}, PlotLegends → ({"Re[G̅]", Table[Gn, {n, 0, 5}]} // Flatten),
 FrameLabel → {λ, G}, PlotTheme → "Scientific", PlotRange → {1/2, 3/2},
 PlotStyle → ({Automatic, Dotted, Dashing[{Small, Medium}],
    Dashing[{0, Small, Tiny}], Dashing[Large], DotDashed}), LabelStyle → Medium]

(*Export["G-pert-series.pdf",%105]*)
```

Relative error shows that the series is asymptotic, not convergent:

```
ListLogPlot[Table[{n, Abs[Gpert[n]/Gexact - 1 /. {m → 1, λ → 1/4}]}, {n, 0, 20}] // N,
 Filling → Bottom, PlotRange → All,
 AxesLabel → {"n", "|(Gn-G̅)/G̅|"}, LabelStyle → Medium]

(*Export["G-pert-asympt-series.pdf",%107]*)
```

Pade Approximants

```
Pade[n_] := Pade[n] = PadeApproximant[Gexact, {λ, 0, {n, n + 1}}]

Table[{n, Pade[n]}, {n, 0, 4}] // TableForm

Plot[{Re[Gexact], Table[Pade[n], {n, 0, 2}]} /. m → 1 // Evaluate, {λ, -3, 3},
 PlotLegends → ({"Re[G̅]", Table[Subsuperscript[Pade, n + 1, n], {n, 0, 2}]} // Flatten),
 FrameLabel → {λ, G}, PlotTheme → "Scientific", PlotRange → {0, 2},
 PlotStyle → {Automatic, Dashed, Dotted, DotDashed}, LabelStyle → Medium]
```

```
(*Export["Pade-approx.pdf",%109]*)

discPade[n_, ϵ_] := (Pade[n] /. λ → λ + ⅈ ϵ) - (Pade[n] /. λ → λ - ⅈ ϵ) //
   FullSimplify[#, Assumptions → {m > 0, λ < 0, ϵ > 0}] &

discPade[0, ϵ]
```

Spectral function:

```
σPade[n_, ϵ_] := σPade[n, ϵ] = (Im[discPade[n, ϵ]] /. λ → -λ)/(-2 π) HeavisideTheta[λ]

σPade[0, 0.0001] /. m → 1

Plot[{σexact, Table[σPade[n, 0.00001], {n, 0, 3}]} /. m → 1 // Evaluate,
 {λ, 0, 10}, FrameLabel → {λ, σ}, PlotTheme → "Scientific",
 PlotLegends → ({"σ", Table[σ[Subsuperscript[Pade, n + 1, n]], {n, 0, 3}]} // Flatten),
 PlotRange → {0, 1/3}, PlotStyle → {Automatic, Dashed, Dotted, DotDashed},
 PlotPoints → 100, LabelStyle → Medium]

(*Export["sigma-Pade.pdf",%111]*)
```

Borel Pade

```
ClearAll[BorelSeriesCoeff, BorelPade, BorelTransf]

BorelSeriesCoeff[n_] := BorelSeriesCoeff[n] =
  (SeriesCoefficient[Gexact, {λ, 0, n}, Assumptions → {m > 0, λ > 0}] 1/n!)

BorelTransf[n_] := BorelTransf[n] = Sum[BorelSeriesCoeff[k] λ^k x^k, {k, 0, n}]

BorelPade[a_, b_] :=
 BorelPade[a, b] = PadeApproximant[BorelTransf[a + b], {x, 0, {a, b}}]

Table[BorelPade[a, b], {a, 3}, {b, 3}] // Simplify

BPG[a_, b_] :=
 BPG[a, b] = LaplaceTransform[BorelPade[a, b], x, s] /. s → 1 // Simplify //
   FullSimplify[#, Assumptions → {λ > 0, m > 0}] &

Table[BPG[a, 1], {a, 0, 3}]

ClearAll[NBPG, NPGs]

NBPG[a_, b_] := NBPG[a, b] =
  Table[{λ, LaplaceTransform[BorelPade[a, b], x, s] /. s → 1 /. m → 1}, {λ, 0, 5, 0.1}]
```

```
(*NBPG[1,1]=
  Table[{λ,NIntegrate[Exp[-x]BorelPade[1,2]/.m→1,{x,0,∞},WorkingPrecision→30]},
   {λ,0,5,0.1}];*)
NBPG[2, 1] = Table[{λ, NIntegrate[Exp[-x] BorelPade[2, 2] /. m → 1,
      {x, 0, ∞}, WorkingPrecision → 45]}, {λ, 0, 5, 0.1}];
NBPG[3, 1] = Table[{λ, NIntegrate[Exp[-x] BorelPade[3, 2] /. m → 1,
      {x, 0, ∞}, WorkingPrecision → 30]}, {λ, 0, 5, 0.1}];
NBPG[1, 2] = Table[{λ, NIntegrate[Exp[-x] BorelPade[1, 2] /. m → 1,
      {x, 0, ∞}, WorkingPrecision → 30]}, {λ, 0, 5, 0.1}];
NBPG[2, 2] = Table[{λ, NIntegrate[Exp[-x] BorelPade[2, 2] /. m → 1,
      {x, 0, ∞}, WorkingPrecision → 45]}, {λ, 0, 5, 0.1}];
NBPG[3, 2] = Table[{λ, NIntegrate[Exp[-x] BorelPade[3, 2] /. m → 1,
      {x, 0, ∞}, WorkingPrecision → 30]}, {λ, 0, 5, 0.1}];
NBPG[1, 3] = Table[{λ, NIntegrate[Exp[-x] BorelPade[1, 3] /. m → 1,
      {x, 0, ∞}, WorkingPrecision → 30]}, {λ, 0, 5, 0.1}];
NBPG[2, 3] = Table[{λ, NIntegrate[Exp[-x] BorelPade[2, 3] /. m → 1,
      {x, 0, ∞}, WorkingPrecision → 30]}, {λ, 0, 5, 0.1}];
NBPG[3, 3] = Table[{λ, NIntegrate[Exp[-x] BorelPade[3, 3] /. m → 1,
      {x, 0, ∞}, WorkingPrecision → 30]}, {λ, 0, 5, 0.1}];
```

```
(*Table[NBPG[a,b],{a,3},{b,3}]*)
```

```
NPGs = Table[{a, b, NBPG[a, b]}, {a, 3}, {b, 3}];
```

```
plotdata = ({{#[[ ;; , 1]], #[[ ;; , 2]] / Thread[Gexact /. m → 1 /. λ → #[[ ;; , 1]]]} // Transpose} & /@
      Flatten[NPGs, 1][[ ;; , 3]])[[ ;; , 1, ;; , ;;]];
```

```
Show[Plot[1, {λ, 0, 5}, PlotRange → {0.9, 1.06}, AxesLabel → {λ, "Gapprox / Ḡ"},
  PlotStyle → {Black, Thick}, PlotLegends → {"Ḡ"}], ListPlot[plotdata, PlotLegends →
   ({Subsuperscript["Borel-Pade", ToString[#[[2]]], ToString[#[[1]]]] & /@
       Flatten[NPGs, 1][[ ;; , 1 ;; 2]]} // Flatten),
  PlotMarkers → {Automatic, Medium}], LabelStyle → Medium]
```

```
(*Export["Gbar-PadeBorel.pdf",%132]*)
```

```
BPσs = Table[{a, b, InverseLaplaceTransform[BorelPade[a, b], x, v] /. v → 1},
   {a, 3}, {b, 3}] // FullSimplify[#, Assumptions → λ > 0] &
```

```
Subsuperscript["Borel-Pade", ToString[#[[2]]], ToString[#[[1]]]] & /@
 Flatten[BPσs, 1][[ ;; , 1 ;; 2]]
```

```
Plot[Re@{1, Flatten[BPσs, 1][[ ;; , 3]] / σexact /. m → 1} // Flatten // Evaluate,
 {λ, 0, 20}, PlotLegends →
  ({σ, σ[Subsuperscript["Borel-Pade", ToString[#[[2]]], ToString[#[[1]]]]] & /@
     Flatten[BPσs, 1][[ ;; , 1 ;; 2]]} // Flatten), AxesLabel → {λ, "σapprox / σexact"},
 PlotRange → {-1, 3}, PlotStyle → ({{Automatic, Thick}, Dotted, Dashing[Large],
    DotDashed, Dashing[{Small, Medium}], Dashing[{0, Small, Tiny}],
    Dashing[{Medium, Medium, 0}], DotDashed, {Dotted, Thick}}), LabelStyle → Medium]
```

```
(*Export["sigma-PadeBorel.pdf",%134]*)
```

```
Plot[{σexact,
   (HeavisideTheta[λ] InverseLaplaceTransform[BorelPade[1, 1], x, v] /. v → 1)} /.
  m → 1 // Evaluate, {λ, 0, 10}, PlotStyle → {DotDashed, Dashed}]
```

```
FindRoot[Re[BPG[0, 1]] /. m → 1, {λ, -5}]
```

```
BPG[0, 1] /. m → 1 /. {λ → -5.36902}
```

2 PI Approximations

```
Clear[Γ2, G]
```

$$\texttt{Γ2[G\_] := Γ2[G]} = \frac{1}{2}\,\texttt{Log}\left[G^{-1}\right] + \frac{1}{2}\,m^2\,G + \texttt{γ2[G]}$$

To determine $\gamma 2[G]$ we expand the left and right hand sides of the G equation of motion to $O(G^n)$:

$$\texttt{GeomExpansion[n\_] := Module}\left[\left\{\texttt{γ2 = Sum}\left[\texttt{γ[i]}\,\left(\lambda\,G^2\right)^i,\ \{i,1,n\}\right]\right\},\ \texttt{Series}\left[\left\{\texttt{Gexact},\ \frac{1}{m^2 + 2\,\texttt{D[γ2, G]}}\ \texttt{/. G → Gexact}\right\},\ \{\lambda,0,n\}\right]\ \texttt{// Normal}\right]$$

and match coefficients:

```
nthMatchingEqs[n_] := Thread[CoefficientList[GeomExpansion[n][[1]], λ] ==
   CoefficientList[GeomExpansion[n][[2]], λ]]
```

```
GeomExpansion[3]
```

```
γ2coeffs = Solve[nthMatchingEqs[20], Table[γ[i], {i, 20}]][[1]];
```

$$\texttt{Sum}\left[\texttt{γ[i]}\,\left(\lambda\,G^2\right)^i,\ \{i,1,20\}\right]\ \texttt{/. γ2coeffs}$$

The series has super-exponential coefficients, aymptotically the same as perturbation theory up to a constant factor:

$$\texttt{Table}\left[\left\{i,\ \frac{\texttt{γ[i]}}{.03\,\left(\frac{-2}{3}\right)^{i-1}\,(i-1)!}\right\},\ \{i,20\}\right]\ \texttt{/. γ2coeffs // TableForm // N}$$

```
Clear[Gsolns]
```

```
Gsolns[n_] := Gsolns[n] =
  G /. Solve[0 == D[1/2 Log[G^-1] + 1/2 m^2 G + Sum[γ[i] (λ G^2)^i, {i, 1, n}], G] /. γ2coeffs, G] //
   FullSimplify[#, Assumptions → {m > 0, λ > 0}] &

Gsolns[1]

Gsolns[3]

Series[Gsolns[1], {λ, 0, 1}] // FullSimplify[#, Assumptions → {m > 0, λ > 0}] &

g2loop = Gsolns[1][[2]]

Plot[{Gexact, g2loop, Gsolns[2][[2]]} /. m → 1 // Evaluate, {λ, -5, 5}, PlotLegends →
  {G̅, Subsuperscript[G, 1, "(2)"], Subsuperscript[G, 2, "(2)"]}, PlotRange → {0, 2}]

Series[(Gsolns[2][[1]] /. λ → λ + ⅈ ϵ) - (Gsolns[2][[1]] /. λ → λ - ⅈ ϵ) /. m → 1,
 {ϵ, 0, 0}, Assumptions → {λ < -5}]

(-1/(2 π ⅈ) Series[(g2loop /. λ → λ + ⅈ ϵ) - (g2loop /. λ → λ - ⅈ ϵ) /. m → 1,
      {ϵ, 0, 0}, Assumptions → {λ < -1/2}] //
    FullSimplify[#, Assumptions → {λ < -1/2}] & // Normal) /. λ → -λ

σ2 = Sqrt[2 λ - m^4]/(π λ) HeavisideTheta[λ - m^4/2]

Plot[{σexact, σ2} /. m → 1 // Evaluate, {λ, 0, 10}, FrameLabel → {λ, σ},
 PlotTheme → "Scientific", PlotLegends → {"σ", "σ(1)"}, PlotRange → {0, 1/3},
 PlotStyle → {Automatic, Dashed}, PlotPoints → 50, LabelStyle → Medium]

(*Export["sigma-2pi-2loop.pdf",%137]*)

Plot[Re /@ {Gexact, Gpert[1], PadeApproximant[Gpert[1], {λ, 0, {0, 1}}],
    BPG[0, 1], g2loop} /. m → 1 // Evaluate, {λ, -6, 6}, PlotRange → {-.5, 2.5},
 PlotLegends → {"Exact", "1st Order Pert.", Subsuperscript["Pade", 1, 0],
   Subsuperscript["Borel-Pade", 1, 0], "1st Order 2PI"}, AxesLabel → {λ, "Re[G̅]"},
 PlotStyle → {Thick, Dashed, Dashing[{Small, Medium, Large}], Thin, DotDashed},
 LabelStyle → Medium]

(*Export["Gbar-all-first-order-approxs.pdf",%139]*)
```